
Analisi del filmato Flash rappresentante il tiro di un cannone   (filmato “Cannone”) 
Cercherò di analizzare il filmato che convenzionalmente denomineremo “Cannone”.

Il filmato, già nella sua prima versione, rappresenta un mini gioco, in cui tramite il tiro simulato di un cannone si vuole colpire un obbiettivo (in questo caso un carro armato).

A livello di stage non sono stati introdotti molteplici livelli.  Solo due livelli sono stati utilizzati, uno per le azioni e l’altro per la scena grafica.
Il filmato sulla time line principale funziona grazie alla solita tecnica dei tre fotogrammi.

Il primo contiene le inizializzazioni, il secondo il codice che effettua il controllo di varie operazioni e deve essere richiamato ripetutamente, il terzo solo il salto al secondo fotogramma.

Gli elementi fondamentali di gioco sono già disposti sullo stage e per farli apparire o scomparire si opera tramite la proprietà _visible, comune a tutti i movieclip.  Infatti, questi elementi sono tutti movieclip, quindi il carro, il cannone (in cui solo la canna è un movie clip, mentre il resto è parte  mantenuta statica), l’esplosione, la palla di cannone.

La palla in particolare è un movieclip che renderemo mobile, atto a “colpire” altri movieclip (carro armato).   

Nel primo fotogramma abbiamo varie inizializzazioni, in particolare vengono resi invisibili tutta una serie di elementi, che non avrebbe senso o non sarebbe opportuno appaiano subito.  In particolare vengono resi invisibili la palla (che si trova comunque in una posizione standard davanti alla bocca del cannone) e l’esplosione:
// rende invisibili due movieclip gia' sullo stage

palla._visible = false;

esplosione._visible = false;
Ed anche i testi della legenda (uso dei pulsanti nel gioco), che si è programmato non far vedere in modo permanente):
tst1._visible = false;

tst2._visible = false;

tst3._visible = false;

tst4._visible = false;

Inoltre vi è una serie di calcoli.  Uno di questi serve a posizionare il carro armato la prima volta sullo stage in una posizione casuale, ma opportuna.  Il calcolo infatti è incrementale rispetto alla posizione della canna del cannone canna._x, prevede che la posizione x del carro però dovrà essere almeno a 200 pixel oltre la canna (ossia alla sua destra), come minimo, lasciando un margine di variabilità in un residuo range di 300 pixel (ossia di questi 300 potremmo averli tutti e 300, o nessuno, o 140, o…).  Il carro è posizionato, diciamo così, semicasualmente:
// posizionamento casuale del carro armato...

carro._x = canna._x+200+Math.random()*300;

Inoltre bisogna tenere presente un'altra cosa.  La palla già solo per motivi estetici, non può partire da una parte qualunque dello schermo.  Essa deve uscire dalla bocca del cannone con precisione e avere una velocità orientata in modo congruente.  Il cannone ovviamente per realizzare il gioco è regolabile nella sua inclinazione (non nella sua potenza di tiro). Precisamente è prevista la possibilità di regolare la sua inclinazione da 0° a 70°.   Questa regolazione corrisponde a un reale movimento di una sua parte (la canna) che ruota intorno ad un opportuno centro di rotazione, opportunamente spostato all’indietro.  Il resto della figura, seppur simile nello stile e nel colore è solo un elemento grafico fisso.   
In base alla inclinazione della canna quindi deve variare la posizione della palla.  Vi è però da considerare che essa nelle varie posizioni di partenza che può assumere mantiene sempre una distanza costante dalla bocca della canna stessa, o meglio ancora dal centro di rotazione della canna (centro che fa anche da riferimento per le coordinate della canna sullo stage).
Per calcolare quindi questa distanza si applicano le seguenti formule:
// calcola la distanza iniziale (standard) della palla

// la punto di rotazione del cannone

deltax = palla._x-canna._x;

deltay = palla._y-canna._y;

dist_palla = Math.sqrt(Math.pow(deltax, 2)+Math.pow(deltay, 2));

in pratica si rileva la distanza tra la palla e la canna (nel suo punto di rotazione) sull’asse x, poi si fà la stessa cosa su quello y, tutto questo grazie al fatto che le posizioni sugli assi dei due elementi sono noti, e infine si applica il teorema di Pitagora per trovare la distanza effettiva, ossia l’ipotenusa del triangolo.
Alcune istruzioni servono per fissare dati che vengono usati nel fotogramma 2, quali:

// velocita' iniziale della palla e accelerazione  di "gravita'"

v = 10;

g = 0.2;

che fissano velocità iniziale ed accelerazione di gravità.  Questa è assolutamente necessaria, altrimenti la palla proseguirebbe diritta fuori dallo schermo (per certi giochi, supponendo la velocità molto alta e l’accelerazione di gravità piuttosto esigua rispetto a tale valore), questo potrebbe essere verosimile e utile, ma non nel nostro caso.

Il gioco ha poi dei suoni al suo interno.  Come sono stati inseriti ?  In primo luogo sono stati scelti alcuni file (mi pare i miei fossero dei .wav).  Poi si è effettuata l’importazione del file audio con File / Importa…  questo menù apre una finestra che permette di importare praticamente tutti i formati audio e video esistenti, oltre che le immagini.  Importato il file (nella libreria), realizzo su di esso una operazione di concatenazione (del tutto analoga a quella realizzata con i movieclip  per essere usata con l’istruzione attachMovie) che permette di agganciare l’audio all’interno del listato. La prima cosa da fare a questo proposito è degli oggetti Sound opportuni che tramite attachSound (notare il nome simile ad attachMovie), vengono abbinati ad un particolare file audio della libreria. 

// suoni vari gioco....

SndCann = new Sound(_root);

SndCann.attachSound("audio_cann");

…..
A questo punto non resta altro che comandare la partenza dell’audio che avviene tramite il metodo start().
Ora tutto è pronto per lanciare l’esecuzione effettiva del nostro filmato.  Questo filmato come tutti quelli di questo tipo, ossia con un controllo di tipo programmato sui movimenti, si basa su pochi fotogrammi continuamente ripetuti.  Il secondo fotogramma viene infatti richiamato ogni sesto di secondo, ossia 6 volte al secondo (bisogna infatti tenere conto che il terzo fotogramma occupa anch’esso un suo tempo di 1/12 di secondo, anche se si tratta solo di un salto all’indietro).
Una delle funzioni di tale fotogramma è controllare di continuo se qualcuno dei tasti del gioco sia premuto. I tasti sono C per far comparire il carro armato, spazio per sparare e freccia in alto o in basso per muovere la canna del cannone.  In particolare quest’ ultimo movimento è controllato dal codice:

// rotazione canna cannone tramite tasti UP e DOWN (freccia alto e basso)   ------

if (Key.isDown(Key.UP) && (canna._rotation>-70)) {


canna._rotation = canna._rotation-1;

}

if (Key.isDown(Key.DOWN) && (canna._rotation<0)) {


canna._rotation = canna._rotation+1;

}

if (canna._rotation>0) {


canna._rotation = 0;

}

// -------------------------------------------------------------------------------

Come è possibile vedere per il rilevamento dei tasti viene usato l’oggetto Key.  Esso è un oggetto assai importante.  In particolare Key.isDown(…) controlla se in quel preciso momento il tasto con quel codice (codice del tasto) è premuto.  Gli if selezionano il codice da eseguire nel caso il tasto suddetto risulti premuto, altrimenti tale codice viene ignorato.   In realtà queste condizioni sopra sono composite esse controllano la pressione di un tasto, ma due di esse controllano anche il punto di rotazione della canna del cannone.  Si noti la presenza dell’operatore AND (&& nella forma C, JavaScript, ActionScript ecc.) che permette di combinare le due condizioni.  La canna ruota se è premuto il tasto giusto E se non vengono [image: image4.emf]

superati i rispettivi limiti di inclinazione.  Inseriamo in questo modo quindi ai valori in un range possibile di variazione (quello appunto da 0° a 70°).  Si noti tra l’altro, che essendo considerato il piano cartesiano rovesciato rispetto al normale, angoli che a noi paiono positivi, per Flash sono calcolati come negativi:
Le variazioni avvengono 1 grado alla volta.
Nel caso l’utente faccia “fuoco” con la pressione del tasto spazio, ecco il codice che viene attivato:

// operazione di fuoco (abbinata al tasto SPACE - spazio)

if (Key.isDown(Key.SPACE)) {


// se NON c'e' gia' un colpo in aria...


if (!palla._visible) {



SndCann.start();



palla._visible = true;



// la posizone di partenza della palla dipende dalla posizione



// della canna del cannone e va' calcolata



palla._x = canna._x + (dist_palla * Math.cos(grad_to_rad(-canna._rotation)));



palla._y = canna._y - (dist_palla * Math.sin(grad_to_rad(-canna._rotation)));



// calcolo vettore velocità iniziale palla di cannone



// (dipendendente dall'angolo di tiro)



vx = v * Math.cos(grad_to_rad(-canna._rotation));



vy = - v * Math.sin(grad_to_rad(-canna._rotation));


} 

}

Si noti intanto subito che la condizione dell’if anche in questo caso è composita.   Essa è resa composita in maniera diversa rispetto a prima, infatti non esiste un operatore &&, ma sono stati messi in cascata due if.  In pratica per eseguire questo codice tutti e due gli if indicati devono essere verificati. Se uno dei due non è vero allora il codice non viene eseguito (ciò corrisponde, nei fatti, ad aver usato l’operatore &&).
La seconda condizione per eseguire il codice è che la palla non sia visibile.  In pratica il si vuole che il colpo arrivi a terra o a segno prima di poterne sparare un altro.  Questo è ovvio data la struttura del filmato.  Esiste una sola palla di cannone !!  Quindi se non ricorriamo a creare dinamicamente ulteriori movieclip è chiaro che non possiamo avere due palle contemporaneamente sullo stage.  In questo caso si è scelto di non far partire una seconda cannonata prima che la prima abbia avuto in qualche modo effetto.  Questo è prodotto dal secondo if   if (! palla.visible)  ove ! è l’operatore NOT. Ossia detto in italiano: se la palla non è visibile allora…. (e si và ad eseguire il codice susseguente).

Partendo la cannonata  se la palla non era visibile lo deve subito diventare.  Ma dove ?  In un punto qualunque ?  Sicuramente no.  Ecco quindi che due opportune istruzioni palla._x = …. e palla._y =….. calcolano la posizione opportuna dove far apparire la palla (come punto iniziale ovviamente, poi la palla si muoverà).
Tale punto è calcolato in base all’inclinazione della canna del cannone e ha come dato da far rimanere costante la distanza tra la palla ed il centro (di rotazione) della canna.

Si proiettta tale distanza (sempre uguale) su x e y, a seconda dell’angolo di cui la canna è inclinata, con delle semplici formule trigonometriche, a voi note.  Si somma quanto trovato con la posizione del centro della canna.  Infine viene calcolata anche la direzione della velocità (il suo modulo, ossia il suo valore, lo sappiamo, è stato definito in v nel primo fotogramma).

Le formule per questa operazione sono simili a quelle usate poco sopra nel listato. Si ha un valore orientato v, che deve essere opportunamente suddiviso in direzione x e y a seconda dell’angolo di inclinazione del cannone.  Questo il significato delle ultime due righe.  In tal modo la palla appare (e parte) dal punto giusto e procede con la sua velocità nella opportuna direzione (la stessa velocità direzionata inopportunamente potrebbe portare a risultati assurdi e ridicoli).
Il fotogramma 2 si deve fare ovviamente carico anche di produrre il movimento della palla in volo e questo avviene con le seguenti linee di codice:

// calcolo moto della palla di cannone in volo

if (palla._visible) 

{


// implementa il movimento della palla di cannone a


// velocita' vx e vy su ciascuno dei due assi


palla._x = palla._x + vx;


palla._y = palla._y + vy;


vy = vy + g;
// calcolo effetto della gravità (accelerazione)

}
In pratica se la palla è visibile, essa deve muoversi (avete mai visto una palla di cannone ferma per aria ??), e quindi ecco sopra il codice opportuno.  Anche questo viene rieseguito più e più volte.  La palla ad ogni passaggio cambia la sua posizione, mutando (ossia incrementando) le sue coordinate x e y.  Queste coordinate mutano in funzione delle rispettive velocità vx e vy.  In generale infatti è vero che s = s0 + v*t; in questo caso x = x0 + vx * t, ma siccome consideriamo il nostro ciclo sul fotogramma 2 un tempo unitario (una unità di tempo), allora si ha x = x0 + vx, come infatti è scritto anche nel programma stesso.
Inoltre vi è da notare che esiste una variazione anche della velocità, in particolare della velocità vy.  Infatti questo moto è quello che si dice un moto “uniformemente accelerato”, ossia sottoposto ad una accelerazione costante (quella di gravità) rispetto ad una delle sue componenti (componente y).
Viene quindi variata, istante per istante la velocità vy (che infine si inverte e il proietto dopo aver salito, cade verso il basso).  

Si evidenzia che non abbiamo tenuto conto della resistenza dell’aria, nel modello balistico creato con questo piccolo programma.
Via via la palla procede nel suo tragitto.  Chiaramente verso terra.  Alla fine due sono le possibilità per la palla di cannone.  Collidere col terreno oppure collidere con il carro armato (centro).  Consideriamo il codice che gestisce la prima ipotesi.  Vi è intanto da dire che il terreno è anch’esso un movieclip.  Se non lo fosse non potremmo gestire la sua collisione con la palla.  Le collisioni vengono rilevate tra movieclip:

// gestisce l'eventale collisione tra la palla di cannone

// ed il terreno (colpo a vuoto)....

if (palla.hitTest(terreno) && palla._visible) 

{


palla._visible = false;


// l'esplosione viene posizionata dove prima era la palla...


esplosione._x = palla._x;


esplosione._y = palla._y;


esplosione._visible = true;


esplosione._alpha = 100;


count_expl = 0;


SndBoom.start();       // esplosione proiettile

}
Intanto vi è da notare che la condizione di questa parte di codice controlla se la palla collide con il terreno (movieclip rappresentante il terreno con una linea nera spessa), e anche se la palla è visibile.  Subito dopo infatti lo stesso codice rende la palla non più visibile (la bomba infatti esplode),  e quindi questa parte di codice sicuramente non sarà eseguita di nuovo, fino a quando un’altra palla non impatterà sul terreno.

Il codice poi inserisce il movieclip esplosione (costituito da una sola immagine statica), nella esatta posizione di dove si trovava la palla.  Si potrebbe abbellire il filmato con esplosioni più dinamiche, ossia animate.  Il suono abbinato a questo evento è abbastanza lungo per farlo.

Infine esiste in questo codice una variabile contatore che viene inizializzata (a 0).  Su questa variabile agisce del successivo codice che permette di avere un effetto sfumato, variando l’ _alpha  del movieclip esplosione.  Questo viene fatto per passi successivi controllando quanti passi sono stati fatti sulla variabile count_expl, ossia da quanto l’esplosione si è verificata.  Da più tempo si è verificata e più l’alfa del movieclip esplosione deve esse basso, fino a sfumare del tutto (vedi codice sotto):

// se e' visibile un esplosione essa viene via via

// sfumata con questa parte di codice

if (esplosione._visible) {


count_expl++;


esplosione._alpha = esplosione._alpha - 20;

}

// dopo 4 passaggi si toglie la visibilita' all' esplosione...

if (count_expl == 4)   esplosione._visible = false;
SndBoom.start(), attiva l’esecuzione del suono della esplosione.

E’ importante notare che questo suono ha un esecuzione asincrona ed indipendente dall’esecuzione del filmato.  Esso non è in alcun modo legato a quanto avviene nel filmato una volta lanciato (si pensi per paragone alla funzione setInterval, che agisce con lo stesso principio).
Ciò che succede invece se la palla colpisce il carro armato è descritto da quest’altro frammento di codice:
// parte di codice che gestisce l'eventuale "centro"

// (la palla ha colpito il carro armato)

if (palla.hitTest(carro) && palla._visible) 

{


SndBoom.start();           // esplosione proiettile


palla._visible = false;     


carro.gotoAndPlay(2);     // attiva l'animazione relativa al carro colpito


                          // interna al movieclip "carro"


SndVitt.start();          // suono di vittoria






  

}

Qui i suoni sono due e saranno sovrapposti, proprio perché ognuna delle due esecuzioni procede in modo indipendente ignorando la presenza dell’altra.  Tecniche diverse possono permettere di produrre i due suoni in sequenza e renderli quindi più chiari.  C’e’ da notare che esiste una istruzione che lancia una animazione sul movieclip carro.  Infatti tale movieclip a differenza di esplosione non è costituito solo da un fotogramma, ma bensì da un fotogramma “standard” (carro integro) e poi da un filmato che visualizza la sua distruzione. Questo filmato parte dal fotogramma 2 del movie “carro” e usa tecniche di interpolazione di forma miste ad altre.  Alla fine anche il carro scompare (grazie agli ultimi 3 o 4 fotogrammi di questo filmato, tramite operazioni sull’ _alpha (trasparenza) del filmato, seppur piuttosto veloci (durano più o meno 1/3 di secondo).
Alla fine di questa sequenza (filmato interno sul movieclip “carro”), vi è una istruzione stop.  Una analoga istruzione vi è anche all’inteno del primo fotogramma di “carro”.  Senza tale istruzione “carro” proseguirebbe alla sua autodistruzione automatica (, nel senso che ogni clip “carro” passerebbe subito ai fotogrammi successivi al primo (carro integro), senza attendere l’apposito momento (collisione con il proiettile, o palla,che dir si voglia). 

Qui a lato si può osservare la costituzione piuttosto articolata della barra temporale del movieclip relativo al carrarmato e alla sua esplosione. Si tratta di diverse interpolazioni di forma (movimento di pezzi esplosi) e alcune azioni.  C’e’ da notare che siccome quelle eseguite sono interpolazioni di forma “anomale” (ossia con variazioni nette anche di posizione e rovesciamenti), esse producono nel loro svolgimento delle serie di linee contorte simili a rocchetti di spago arruffato.  Tutto questo alla fine, seppur un po’ empirico e confuso è comunque opportuno per rappresentare una esplosione.  Si aspettano vostre proposte per una eventuale visualizzazione alternativa della esplosione del carro armato.
Infine si nota alla fine del codice del secondo fotogramma una routine (o meglio una funzione), che ha lo scopo di convertire un angolo in gradi in uno in radianti (i radianti sono un altro modo di misurare gli angoli; un angolo giro corrisponde a 6,2831… radianti, equivalenti a 360 gradi; da qui la proporzione segnata e calcolata nel programma; la funzione rende il numero di radianti equivalente ai gradi indicati):
// routine di conversione della misura degli

// angoli da gradi a radianti

function grad_to_rad(gradi) {


return (gradi*6.28318/360);

}

La necessità di questa conversione è data dal fatto che Flash indica gli angoli di rotazione in gradi, ma le sue funzioni sugli angoli (seno, coseno, ecc.) operano in radianti.  Si rende quindi ovviamente necessaria una conversione per permettere i relativi calcoli in modo corretto.
Infine vi è il codice abbinato al pulsante che è preposto a visualizzare la legenda (relativa all’uso dei tasti per far funzionare il programma).  Il codice è concettualmente assai semplice.  I singoli testi sono inclusi in 4 testi dinamici (ossia del tipo che è possibile rendere visibili o invisibili).  In questo modo basta far sì che quando l’apposito pulsante a forma di triangolo è premuto i testi siano tutti visibili, o viceversa che quando il pulsante viene rilasciato essi non siano più visibili:
// mostra la legenda....

on(press)

{

       tst1._visible = true;


tst2._visible = true;


tst3._visible = true;


tst4._visible = true;


}

// nasconde la legenda....

on(release)

{

       tst1._visible = false;


tst2._visible = false;


tst3._visible = false;


tst4._visible = false;


Qui a lato il contenuto della libreria del filmato “cannone”.

